
PWL_Noise_Add
* www.idea2ic.com dsauersanjose@aol.com  4/22/08
*
*     OUT     RLOAD      OUT2    RLOAD2      OUT3    RLOAD3       ____
*      _/\  /\  /\_       _/\  /\  /\_        _/\  /\  /\_       |    |-> V_AVE
*     |   \/  \/   |     |   \/  \/   |      |   \/  \/   | OUT->| RMS|
*    _|__         _|_   _|__         _|_    _|__         _|      |____|-> V_RMS
*   /    \        ///  /    \        ///   /OUT \ B1     //      ____
*  / VPWL \           / VPWL \            /  +   \               |    |-> V_AVE2
*  \      /           \      /            \ OUT2 /         OUT2->| RMS|
*   \____/             \____/              \____/                |____|-> V_RMS2
*     |  PWL_test.txt    |   PWL_test2.txt   |                   ____
*    _|_                _|_                 _|_                  |    |-> V_AVE3
*    ///                ///                 ///            OUT3->| RMS|
*                                                                |____|-> V_RMS3
*
*  VpwlT OUT   0  PWL(       <==  Format for piecewise linear file is as so. 
*  + 0      0.38156547         Random Noise can by a javascript calculator.
*  + 1.00E-06 0.60066694         The step of time is 0.5ms and bandwidth
*  + 2.00E-06 0.361341313        of noise will therefore be a half of 1kHz. 
*  .........etc............
*  + 0.000003 0.944365605        The total measurement time is 500ms which 
*  + 5.16E-04 0.658730205        corresponds to a 2Hz resolution
*  + )
*                                 This text file is located at the following 

*  Users/donsauer/Documents/MacSpice/PWL_test.txt 
*
*   spec     start_f   stop_f   step_f    vector [vector ...]
*   spec  2      2k     2       v(out)        spec 2Hz->2kHz @ 2Hz steps
*  .tran  TSTEP  TSTOP  TSTART TMAX   ?UIC?
*======== ====== ====== ====== ====== ====== ====== ====== ====== ====== ======

.include    PWL_test.txt

.include    PWL_test2.txt
Rload       OUT     0     1k
Rload2      OUT2   0     1k
Rload3      OUT3   0     1k
B1          OUT3   0     v = v(out) + v(out2)
XM_RMS   OUT      V_AVE  V_RMS    M_RMS
XM_RMS2 OUT2    V_AVE2 V_RMS2   M_RMS
XM_RMS3 OUT3    V_AVE3 V_RMS3   M_RMS

.tran     500u   500m     0     500u  UIC

**===See_Two_Random_Waveform_Sum_Together=====

.control
set  outfile =          "Noise_Waveforms.txt"
run
plot      OUT  OUT2 V_RMS V_RMS2
plot      OUT3 V_RMS3

set n
let n = 0
set v
let v = 0
set v2
let v2 = 0
set v3
let v3 = 0

**===Print_Out_The_Results=====
repeat  4420
let n = n+1
let v  = out[n]
let v2 = out2[n]
let v3 = out3[n]
echo                          "$&v   $&v2   $&v3 " >> $outfile
endrepeat

.endc

* ========== .SUBCKT M_RMS V_IN V_AVE V_RMS  ==================
*              ____
*             |    |-> V_AVE
*       V_IN->| RMS|
*             |____|-> V_RMS
*
*
*          Vsum    Rave         Vsum2   Rave2       V_RMS    Rrms
*          __/\  /\  /\_ V_AVE  __/\  /\  /\_ Vave2 __/\  /\  /\_
*        _|_   \/  \/   |     _|_   \/  \/   |    _|_   \/  \/   |
* V_IN->/   \          _|_   /   \          _|_  /   \          _|_
*       \___/Bsum      ___   \___/Bsum2     ___  \___/Brms      ///
*         |             |      |             |     |
*        _|_      Cave _|_    _|_     Cave2 _|_   _|_
*        ///           ///    ///           ///   ///
*

.SUBCKT M_RMS  V_IN  V_AVE V_RMS
Bsum       Vsum      0         V = v(V_IN)
Rave         Vsum      Vave      1
Cave         V_AVE  0         300m
Bsum2     Vsum2     0         V = (v(V_IN)-v(V_AVE))^2
Rave2        Vsum2     Vave2     1
Cave2        Vave2     0         100m
Brms       V_RMS  0         V = v(Vave2)^0.5
Rrms         V_RMS  V0        1k
.ENDS   M_RMS



.end

=====================END_OF_SPICE============================

The total sample period is 200msec(2Hz)
Two 1Vrms noise signals were generated.

Two Noise Waveforms are generated.



The sum of the two is shown above.

To Sanity check the V_RMS values
The wave data is pasted into a javascript

http://www.idea2ic.com/PlayWithJavascript/RMS_DATA_OF_COLUMNS.html



* http://www.idea2ic.com/PlayWithJavascript/RandomPWL2.html 




